
Changelog
    


  
    
  
  v0.26.2


	Introduces warning if plural messages are defined with the same singular
message and conflicting plural messages.
	Improves performance by striping not required metadata when compiling the
Gettext backend.


  
    
  
  v0.26.1


	Address backwards incompatible changes in previous release


  
    
  
  v0.26.0


This release changes the way you use Gettext. We're not crazy: it does so because doing so makes it a lot faster to compile projects that use Gettext.
The changes you have to make to your code are minimal, and the old behavior is deprecated so that you will be guided on how to update.
The reason for this change is that it removes compile-time dependencies from modules that used to import a Gettext backend. In applications such as Phoenix applications, where every view and controller imports the Gettext backend, this change means a lot less compilation when you make translation changes!
Here's the new API. Now, instead of defining a Gettext backend (use Gettext) and then importing that to use its macros, you need to:
	Define a Gettext backend with use Gettext.Backend
	Import and use its macros with use Gettext, backend: MyApp.Gettext.


  
    
  
  Before and After


Before this release, code using Gettext used to look something like this:
defmodule MyApp.Gettext do
  use Gettext, otp_app: :my_app
end

defmodule MyAppWeb.Controller do
  import MyApp.Gettext
end
This creates a compile-time dependency for every module that imports the Gettext backend.
With this release, the above turns into:
defmodule MyApp.Gettext do
  use Gettext.Backend, otp_app: :my_app
end

defmodule MyAppWeb.Controller do
  use Gettext, backend: MyApp.Gettext
end
We are also updating Phoenix generators to use the new API.
If you update Gettext and still use use Gettext, otp_app: :my_app to define a backend, Gettext will emit a warning now.

  
    
  
  Migration with Igniter


If your project is using igniter, you can run
mix igniter.update_gettext
to automatically migrate to the new API.

  
    
  
  Detailed Changelog


This is a detailed list of the new things introduced in this release:
	Add Gettext.Macros, which contains all the macros you know and love (*gettext). It also contains *gettext_with_backend variants to explicitly pass a backend at compile time and keep extraction working.
	Document lgettext/5 and lngettext/7 callbacks in Gettext.Backend. These get generated in every Gettext backend.
	Add the Gettext.domain/0 type.


  
    
  
  v0.25.0


	Run merging for mix gettext.extract's POT files even if they are unchanged.
	Allow Expo 1.0+.


  
    
  
  v0.24.0


	Handle singular and plural messages with the same msgid as the same
message.
This change produces a Expo.PO.DuplicateMessagesError if you already have
messages with the same singular msgid. This can be solved by calling the
expo.msguniq mix task on your .po file:
mix expo.msguniq \
  priv/gettext/LOCALE/LC_MESSAGES/DOMAIN.po \
  --output-file priv/gettext/LOCALE/LC_MESSAGES/DOMAIN.po




  
    
  
  v0.23.1


	Use the Hex version of the excoveralls dependency.


  
    
  
  v0.23.0


	Add the :custom_flags_to_keep Gettext option.


  
    
  
  v0.22.3


	Fix a bug with extracting translations in Elixir 1.15.0+.


  
    
  
  v0.22.2


	Use Code.ensure_compiled/1 instead of Code.ensure_loaded/1 for Elixir < 1.12 compatibility.
	Ensure all modules are properly loaded for mix gettext.merge.
	Fix a "protected" check when extracting translations.


  
    
  
  v0.22.1


	Put correct Plural-Forms header on gettext.merge for the first time.
	Fix extractor crash in case of conflicting backends.
	Fix to use the correct plural forms for multiple languages.
	Update expo to ~> 0.4.0 to fix issues with empty msgstr.


  
    
  
  v0.22.0


	Deprecate (with a warning) the --plural-forms CLI option and the :plural_forms option in favor of --plural-forms-header and :plural_forms_header.
	Supply the Plural-Forms header to Gettext.Plural callbacks.
	Bump Expo requirement to ~> 0.3.0.
	Add the types:	Gettext.Interpolation.bindings/0
	Gettext.Error.t/0
	Gettext.Plural.locale/0
	Gettext.Plural.pluralization_context/0
	Gettext.Plural.plural_info/0


	Add the optional callbacks Gettext.Plural.init/1 and Gettext.Plural.plural_forms_header/1.


  
    
  
  Bug fixes


	Fix --check-up-to-date with msgids split in different ways.
	Don't write the same file more than once in references when using write_reference_line_numbers: false.


  
    
  
  v0.21.0



  
    
  
  New features and improvements


	Bump Elixir requirement to 1.11+.

	Extract parsing and dumping of PO/POT files to the
expo library, and start depending
on that.

	Support marking messages as obsolete with the new :on_obsolete Gettext
configuration option.

	Add the :write_reference_line_numbers Gettext configuration option.

	Save the previous messages when there's a fuzzy match, with the new
:store_previous_message_on_fuzzy_match Gettext configuration option.

	Change :sort_by_msgid to accept false, :case_sensitive, or
:case_insensitive and deprecate the true value.



  
    
  
  Bug fixes


	Sort messages independent of line splits when dumping PO files.


  
    
  
  v0.20.0


	Allow gettext_comment to be invoked multiple times
	Dump flags after references in PO files
	Deprecate compile.gettext in favor of __mix_recompile__?


  
    
  
  Backwards incompatible changes


	handle_missing_translation(locale, domain, msgid, bindings) callback
signature was changed to handle_missing_translation(locale, domain, msgctxt, msgid, bindings) (it receives a new argument called msgctxt)

	handle_missing_plural_translation(locale, domain, msgid, msgid_plural, n, bindings) callback signature was changed to
handle_missing_plural_translation(locale, domain, msgctxt, msgid, msgid_plural, n, bindings) (it receives a new argument called msgctxt)



  
    
  
  v0.19.1


	Fix warnings on Elixir v1.14+
	Rename ex-autogen to elixir-autogen and make sure elixir-autogen is
added to existing messages


  
    
  
  v0.19.0


	Remove the :one_module_per_locale option in favor of  :split_module_by
and :split_module_compilation
	Make Gettext.dngettext/6 bindings argument optional (effectively
introducing Gettext.dngettext/5)
	Preserve the fuzzy message flag when merging
	Add the --check-unextracted flag to mix gettext.extract, which is useful
in CI and similar
	Place each message reference on its own line in extracted PO files
	Make the interpolation module customizable via the :interpolation
configuration option
	Use a different flag to detect autogenerated messages (ex-autogen)
	Update gettext.extract to correctly extract on recompilation for Elixir
1.13+


  
    
  
  v0.18.2


	Allow plural forms to be set for the :gettext application
	Use Application.compile_env/3 if available


  
    
  
  v0.18.1


	Allow default domain to be configurable
	Improve parallelism when compiling modules


  
    
  
  v0.18.0


	Allow sorting strings by msgid
	Add :allowed_locales to restrict the locales bundled in the backend


  
    
  
  v0.17.4


	Do not change the return types of *_noop macros (regression in v0.17.2 and
v0.17.3)
	Fix dialyzer warnings


  
    
  
  v0.17.3


	Add lgettext/4 back which was removed in v0.17.2 - note lgettext/4 is
private API and may be removed in future once again


  
    
  
  v0.17.2


	Support pgettext
	Consider extracted comments when merging templates during extraction


  
    
  
  v0.17.1


	Store the msgctxt value in message and dump it when dumping
messages
	Fix a bug when dumping references
	Improve code generation
	Preserve whitespace in message flags


  
    
  
  v0.17.0


	Require Elixir 1.6 and later
	Add stats reporting when merging PO files


  
    
  
  v0.16.1


	Optimize default locale lookup


  
    
  
  v0.16.0


	Fix bugs related to expanding arguments to Gettext macros
	Fix a bug where you couldn't have filenames with colons in them in reference comments
	Add handle_missing_translation/4 and handle_missing_plural_translation/6 callbacks to Gettext backends
	Fix a bug in mix gettext.extract, which was ignoring the --merge option


  
    
  
  v0.15.0


	Generate correct plural forms when dumping new messages in PO files
	Fix a bug where we were losing translator comments for fuzzy-merged
messages
	Don't make an exact match when merging prevent later fuzzy matches
	Allow multiple messages to fuzzy-match against the same message when
merging
	Bump the Elixir requirement to v1.4 and on


  
    
  
  v0.14.1


	Copy flags from existing messages when merging messages


  
    
  
  v0.14.0


	Introduce a global locale (per-process) for all Gettext backends
	Warn when compiling and raise at runtime for missing plural forms
	Separate flags with commas when dumping and parsing .pot files
	Add support for extracted comments via gettext_comment/1
	Require Elixir v1.3 and fix warnings
	Improve compilation time of Gettext backends in roughly 20%
	Add :one_module_per_locale for parallel compilation of backends (requires
Elixir v1.6)
	Use the elixir-format flag to mark autogenerated messages


  
    
  
  v0.13.1


	Fix a bug with Dialyzer specs for the Gettext.Backend.ngettext_noop/2 callback
	Parse msgctxt entries in PO and POT files so that they don't cause syntax
errors, but ignore them in the parsed result


  
    
  
  v0.13.0


	Add the gettext_noop/1, dgettext_noop/2, ngettext_noop/3, and
dngettext_noop/4 macros to Gettext backends. These macros can be used to
mark messages for extractions without translating the given string


  
    
  
  v0.12.2


	Fix a bug where we failed miserably with a "no process" error when
extracting messages without having the :gettext compiler run
	Slightly revisit the indentation of subsequent literal strings in dumped
PO(T) files; before, they were dumped one per line, indented one level more
than the parent message, while now they're indented at the same level as
the parent message


  
    
  
  v0.12.1


	Ensure the Gettext application is started before running Mix tasks


  
    
  
  v0.12.0


	Drop support for Elixir 1.1 and require ~> 1.2
	Add :compiler_po_wildcard to explicitly choose the po files that are
tracked by the compiler
	Allow the developer to configure what happens when there are missing
bindings in the message. The default has been changed to log and return
the incomplete string instead of raising
	Move the configuration for the :gettext application to compile-time config
in project/0 in mix.exs (under the :gettext key, with configuration
options :excluded_refs_from_purging, :compiler_po_wildcard and
:fuzzy_threshold)
	Show the file name in syntax errors when running mix gettext.extract and
mix gettext.merge
	Don't print tokens as Erlang terms in syntax error when running mix gettext.extract and mix gettext.merge
	Allow duplicate interpolation keys
	Raise when the domain is not a binary at compile-time
	Fix many dialyzer warnings
	No longer traverse directories given to gettext.merge recursively (from
now on gettext.merge expect specific locale directories)
	Re enable the "compile" task in mix gettext.extract
	Ensure messages are tracked to the proper child app when using umbrella
apps


  
    
  
  v0.11.0


	Polish so many docs!
	Make an error in Gettext.put_locale/2 clearer
	Pluralize x_Y locales as x, but fail with
Gettext.Plural.UnknownLocaleError for any other unknown locale
	Add a Gettext.Backend behaviour (automatically implemented if a module
calls use Gettext)
	Allow whitelisting of references via the :excluded_refs_from_purging option
in the :gettext application config


  
    
  
  v0.10.0


	Emit warnings when the domain passed to one of the *gettext macros has
slashes in it (as we don't support domains in subdirectories).
	Discard dangling comments when parsing/dumping PO(T) files (dangling
comments are comments that are not followed by a transaction they can be
attached to).
	Updated informative comments for newly generated PO/POT files.


  
    
  
  v0.9.0


	Strip ## comments from POT files when they're being merged into PO files;
these comments are comments meant to be generated by tools or directed at
developers (so they have no use for translators in PO files).
	Add informative comments at the top of newly generated PO/POT files.
	Add Gettext.known_locales/1
	Fix a bug with PO parsing when the PO file starts with a
BOM character (which broke
the parser, now a warning is issued).


  
    
  
  v0.8.0


	Fix a bug with the *gettext macros, which raised an error when given
compile-time strings in the form of ~s/~S sigils.
	Create missing locale directories (for example, en/LC_MESSAGES) when
running the gettext.merge Mix task.
	Fallback to default messages (that is, the msgid) when the msgstr
(or one or more msgstr strings for plural messages) is empty.


  
    
  
  v0.7.0


	When dumping PO files, dump as many references as possible on one line,
wrapping at the 80th column
	Parse multiple references in the same reference comment
	Remove Gettext.locale/0-1 and Gettext.with_locale/2 in favour of
Gettext.get_locale/1, Gettext.put_locale/2, and Gettext.with_locale/3
which now work by setting/getting the locale on a per-backend basis (instead
of a global one)
	Remove the :default_locale config option for the :gettext application in
favour of configuring the :default_locale for backends tied to their
:otp_app (for example, config :my_app, MyApp.Gettext, default_locale: "pt_BR")


  
    
  
  v0.6.1


	Fix a bug with the mix gettext.merge task that was failing in Elixir
v1.1.1 because 0.5 in 0..1 returns false with it


  
    
  
  v0.6.0


	Add a :flags field to the Gettext.PO.Translation and
Gettext.PO.PluralTranslation structs
	Add support for fuzzy matching messages in gettext.merge and
gettext.extract --merge
	Add the :fuzzy_threshold configuration option for the :gettext
application


  
    
  
  v0.5.0


	Initial release



  

    
Gettext 
    



      
The Gettext module provides a
gettext-based API for working with
internationalized applications.

  
    
  
  Basic Overview


When you use Gettext, you replace hardcoded user-facing text like this:
"Hello world"
with calls like this:
gettext("Hello world")
Here, the string "Hello world" serves two purposes:
	It's displayed by default (if no translation is specified in the current
language). This means that, at the very least, switching from a hardcoded
string to a Gettext call is harmless.

	It serves as the message ID to which translations will be mapped.


An example translation workflow is as follows.
First, call mix gettext.extract to extract gettext() calls to .pot
(Portable Object Template)
files, which are the base for all translations. These files are templates, which
means they only contain message IDs, and not actual translated strings. POT files have
entries like this:
#: lib/myapp_web/live/hello_live.html.heex:2
#, elixir-autogen, elixir-format
msgid "Hello world"
msgstr ""
Then, call mix gettext.merge priv/gettext to update all
locale-specific .po (Portable Object) files so that they include this message ID.
Entries in PO files contain translations for their specific locale. For example,
in a PO file for Italian, the entry above would look like this:
#: lib/myapp_web/live/hello_live.html.heex:2
#, elixir-autogen, elixir-format
msgid "Hello world"
msgstr "Ciao mondo"
The English string is the msgid which is used to look up the
correct Italian string.
That's handy, because unlike a generic key like site.greeting (as some
translations systems use), the message ID tells exactly what needs to be
translated. This is easier to work with for translators, for example.
But it raises a question: what if you change the original English string in the code?
Does that break all translations, requiring manual edits everywhere? Not necessarily.
After you run mix gettext.extract again, the next mix gettext.merge can
do fuzzy matching.
So, if you change "Hello world" to "Hello world!", Gettext will see that the new
message ID is similar to an existing msgid, and will do two things:
	It will update the msgid in all .po files to match the new text.

	It will mark those entries as "fuzzy"; this hints that a (probably human)
translator should check whether the Italian translation of this string needs
an update.


The resulting change in the .po file is this (note the "fuzzy" annotation):
#: lib/myapp_web/live/hello_live.html.heex:2
#, elixir-autogen, elixir-format, fuzzy
msgid "Hello world!"
msgstr "Ciao mondo"
This "fuzzy matching" behavior can be configured or disabled, but its
existence makes updating translations to match changes in the base text easier.
The rest of the documentation will cover the Gettext API in detail.

  
    
  
  Gettext API


To use Gettext, you will need a backend module which stores and retrieves
translations from PO files. You can create such a module by using Gettext.Backend:
defmodule MyApp.Gettext do
  use Gettext.Backend, otp_app: :my_app
end
Now, you can import all the necessary translation macros (defined in Gettext.Macros)
into any module by using Gettext:
defmodule MyApp.SomeModule do
  use Gettext, backend: MyApp.Gettext

  def showcase_gettext do
    # Simple message
    gettext("Hello world")

    # Plural message
    ngettext(
      "Here is the string to translate",
      "Here are the strings to translate",
      3
    )

    # Domain-based message
    dgettext("errors", "Here is the error message to translate")

    # Context-based message
    pgettext("email", "Email text to translate")
  end
end
The arguments for the Gettext macros and their order can be derived from
their names. For example, for dpgettext/4
the arguments are: domain, context, msgid, bindings (default to %{}).
Messages are looked up from .po files. In the following sections we will
explore exactly what are those files before we explore the "Gettext API" in
detail.
Recent Updates
Before v0.26.0 of this library, the workflow described in this section
was slightly different. Check out the
changelog for more
details, but the gist is that use Gettext used to define macros in the calling module.
This created heavy compile-time dependencies which would cause slow recompilation
in larger applications.


  
    
  
  Messages


Messages are stored inside PO (Portable Object) files, with a .po
extension. For example, this is a snippet from a PO file:
# This is a comment
msgid "Hello world!"
msgstr "Ciao mondo!"
PO files containing messages for an application must be stored in a
directory (by default it's priv/gettext) that has the following structure:
gettext directory
└─ locale
   └─ LC_MESSAGES
      ├─ domain_1.po
      ├─ domain_2.po
      └─ domain_3.po
Here, locale is the locale of the messages (for example, en_US),
LC_MESSAGES is a fixed directory, and domain_i.po are PO files containing
domain-scoped messages. For more information on domains, check out the
"Domains" section below.
A concrete example of such a directory structure could look like this:
priv/gettext
└─ en_US
|  └─ LC_MESSAGES
|     ├─ default.po
|     └─ errors.po
└─ it
   └─ LC_MESSAGES
      ├─ default.po
      └─ errors.po
By default, Gettext expects messages to be stored under the priv/gettext
directory of an application. This behaviour can be changed by specifying a
:priv option when using Gettext:
# Look for messages in my_app/priv/messages instead of
# my_app/priv/gettext
use Gettext.Backend,
  otp_app: :my_app,
  priv: "priv/messages"
The messages directory specified by the :priv option should be a directory
inside priv/, otherwise some things won't work as expected.

  
    
  
  Locale


At runtime, all gettext-related functions and macros that do not explicitly
take a locale as an argument read the locale from the backend and fall back
to Gettext's default locale.
Gettext.put_locale/1 can be used to change the locale of all backends for
the current Elixir process. That's the preferred mechanism for setting the
locale at runtime. Gettext.put_locale/2 can be used when you want to set the
locale of one specific Gettext backend without affecting other Gettext
backends.
Similarly, Gettext.get_locale/0 gets the locale for all backends in the
current process. Gettext.get_locale/1 gets the locale of a specific backend
for the current process. Check their documentation for more information.
Locales are expressed as strings (like "en" or "fr"); they can be
arbitrary strings as long as they match a directory name. As mentioned above,
the locale is stored per-process (in the process dictionary): this means
that the locale must be set in every new process in order to have the right
locale available for that process. Pay attention to this behaviour, since not
setting the locale will not result in any errors when Gettext.get_locale/0
or Gettext.get_locale/1 are called; the default locale will be
returned instead.
To decide which locale to use, each gettext-related function in a given
backend follows these steps:
	if there is a backend-specific locale for the given backend for this
process (see put_locale/2), use that, otherwise
	if there is a global locale for this process (see put_locale/1), use
that, otherwise
	if there is a backend-specific default locale in the configuration for
that backend's :otp_app (see the "Default locale" section below), use
that, otherwise
	use the default global Gettext locale (see the "Default locale" section
below)


  
    
  
  Default locale


The global Gettext default locale can be configured through the
:default_locale key of the :gettext application:
config :gettext, :default_locale, "fr"
By default the global locale is "en". See also get_locale/0 and
put_locale/1.
If for some reason a backend requires a different :default_locale
than all other backends, you can set the :default_locale inside the
backend configuration, but this approach is generally discouraged as
it makes it hard to track which locale each backend is using:
config :my_app, MyApp.Gettext, default_locale: "fr"

  
    
  
  Gettext API


There are two ways to use Gettext:
	using macros from your own Gettext module, like MyApp.Gettext
	using functions from the Gettext module

These two approaches are different and each one has its own use case.

  
    
  
  Using macros


Each module that calls use Gettext.Backend is usually referred to as a "Gettext
backend", as it implements the Gettext.Backend behaviour. When a module then calls
use Gettext, backend: MyApp.Gettext, all the macros defined in Gettext.Macros
are imported into that module, such as:
	gettext/2
	dgettext/3
	pgettext/3

Using macros is preferred as Gettext is able to automatically sync the
messages in your code with PO files. This, however, imposes a constraint:
arguments passed to any of these macros have to be strings at compile
time. This means that they have to be string literals or something that
expands to a string literal at compile time (for example, a module attribute like
@my_string "foo").
These are all valid uses of the Gettext macros:
Gettext.put_locale(MyApp.Gettext, "it")

use Gettext, backend: MyApp.Gettext

gettext("Hello world")
#=> "Ciao mondo"

@msgid "Hello world"
gettext(@msgid)
#=> "Ciao mondo"
The *gettext macros raise an ArgumentError exception if they receive a
domain, msgctxt, msgid, or msgid_plural that doesn't expand to a string
at compile time:
msgid = "Hello world"
gettext(msgid)
#=> ** (ArgumentError) msgid must be a string literal
Using compile-time strings isn't always possible. For this reason,
the Gettext module provides a set of functions as well.

  
    
  
  Using functions


If compile-time strings cannot be used, the solution is to use the functions
in the Gettext module instead of the macros described above. These functions
perfectly mirror the macro API, but they all expect a Gettext backend module
as the first argument.
defmodule MyApp.Gettext do
  use Gettext.Backend, otp_app: :my_app
end

Gettext.put_locale(MyApp.Gettext, "pt_BR")

msgid = "Hello world"
Gettext.gettext(MyApp.Gettext, msgid)
#=> "Olá mundo"
While using functions from the Gettext module yields the same results as
using macros (with the added benefit of dynamic arguments), all the
compile-time features mentioned in the previous section are lost.

  
    
  
  Domains


The dgettext and dngettext
macros (and their function counterparts) also accept a domain as one
of the arguments. The domain of a message is determined by the name of the
PO file that contains that message. For example, the domain of
messages in the it/LC_MESSAGES/errors.po file is "errors", so those
messages would need to be retrieved with dgettext or dngettext:
dgettext("errors", "Error!")
#=> "Errore!"
When backend gettext, ngettext, or pgettext are used, the backend's
default domain is used (which defaults to "default"). The Gettext
functions accepting a backend (gettext/3, ngettext/5, and pgettext/4)
always use a domain of "default".

  
    
  
  Default Domain


Each backend can be configured with a specific :default_domain
that replaces "default" in gettext/2, pgettext/3, and ngettext/4
for that backend.
defmodule MyApp.Gettext do
  use Gettext.Backend,
    otp_app: :my_app,
    default_domain: "messages"
end

config :my_app, MyApp.Gettext, default_domain: "messages"

  
    
  
  Contexts


The GNU Gettext implementation supports
contexts,
which are a way to contextualize messages. For example, in English, the
word "file" could be used both as a noun as well as a verb. Contexts can be used to
solve similar problems: you could have a imperative_verbs context and a
nouns context as to avoid ambiguity. The functions that handle contexts
have a p in their name (to match the GNU Gettext API), and are pgettext,
dpgettext, pngettext, and dpngettext. The "p" stands for "particular".

  
    
  
  Interpolation


All *gettext functions and macros provided by Gettext support interpolation.
Interpolation keys can be placed in msgids or msgid_plurals with by
enclosing them in %{ and }, like this:
"This is an %{interpolated} string"
Interpolation bindings can be passed as an argument to all of the *gettext
functions/macros. For example, given the following PO file for the "it"
locale:
msgid "Hello, %{name}!"
msgstr "Ciao, %{name}!"
interpolation can be done like follows:
Gettext.put_locale(MyApp.Gettext, "it")
gettext("Hello, %{name}!", name: "Meg")
#=> "Ciao, Meg!"
Interpolation keys that are in a string but not in the provided bindings
result in an exception:
gettext("Hello, %{name}!")
#=> ** (Gettext.MissingBindingsError) ...
Keys that are in the interpolation bindings but that don't occur in the string
are ignored. Interpolations in Gettext are often expanded at compile time,
ensuring a low performance cost when running them at runtime.

  
    
  
  Pluralization


Pluralization in Gettext for Elixir works very similar to how pluralization
works in GNU Gettext. The *ngettext functions/macros accept a msgid, a
msgid_plural, and a count of elements; the right message is chosen based
on the pluralization rule for the given locale.
For example, given the following snippet of PO file for the "it" locale:
msgid "One error"
msgid_plural "%{count} errors"
msgstr[0] "Un errore"
msgstr[1] "%{count} errori"
the ngettext macro can be used like this:
Gettext.put_locale(MyApp.Gettext, "it")
ngettext("One error", "%{count} errors", 3)
#=> "3 errori"
The %{count} interpolation key is a special key since it gets replaced by
the number of elements argument passed to *ngettext, like if the count: 3
key-value pair were in the interpolation bindings. Hence, never pass the
count key in the bindings:
# `count: 4` is ignored here
ngettext("One error", "%{count} errors", 3, count: 4)
#=> "3 errori"
You can specify a "pluralizer" module via the :plural_forms option in the
configuration for each Gettext backend.
defmodule MyApp.Gettext do
  use Gettext.Backend,
    otp_app: :my_app,
    plural_forms: MyApp.PluralForms
end
To learn more about pluralization rules, plural forms and what they mean to
Gettext check the documentation for Gettext.Plural.

  
    
  
  Missing messages


When a message is missing in the specified locale (both with functions and
with macros), the argument is returned:
	in case of calls to gettext/dgettext/pgettext/dpgettext, the msgid argument is returned
as is;
	in case of calls to ngettext/dngettext/pngettext/dpngettext, the msgid argument is
returned in case of a singular value and the msgid_plural is returned in
case of a plural value (following the English pluralization rule).

For example:
Gettext.put_locale(MyApp.Gettext, "foo")
gettext("Hey there")
#=> "Hey there"
ngettext("One error", "%{count} errors", 3)
#=> "3 errors"

  
    
  
  Empty messages


When a msgstr is empty (""), the message is considered missing and the
behaviour described above for missing message is applied. A plural
message is considered to have an empty msgstr if at least one
message in the msgstr is empty.

  
    
  
  Compile-time features


As mentioned above, using the Gettext macros (as opposed to functions) allows
Gettext to operate on those messages at compile-time. This can be used
to extract messages from the source code into POT (Portable Object Template)
files automatically (instead of having to manually add messages to POT files
when they're added to the source code). mix gettext.extract does exactly
this: whenever there are new messages in the source code, running
this task syncs the existing POT files with the changed code base.
Read the documentation for mix gettext.extract for more information
on the extraction process.
POT files are just template files and the messages in them do not
actually contain translated strings. A POT file looks like this:
# The msgstr is empty
msgid "hello, world"
msgstr ""
Whenever a POT file changes, it's likely that developers (or translators) will
want to update the corresponding PO files for each locale. To do that, gettext
provides the gettext.merge Mix task. For example, running:
mix gettext.merge priv/gettext --locale pt_BR
will update all the PO files in priv/gettext/pt_BR/LC_MESSAGES with the new
version of the POT files in priv/gettext. Read more about the merging
process in the documentation for mix gettext.merge.

  
    
  
  Configuration



  
    
  
  :gettext configuration


The :gettext application supports the following configuration options:
	:default_locale - a string which specifies the default global Gettext
locale to use for all backends. See the "Locale" section for more
information on backend-specific, global, and default locales.


  
    
  
  Backend configuration


A Gettext backend supports some options to be configured. These options
can be configured in two ways: either by passing them to use Gettext (hence
at compile time):
defmodule MyApp.Gettext do
  use Gettext.Backend, options
end
or by using Mix configuration, configuring the key corresponding to the
backend in the configuration for your application:
# For example, in config/config.exs
config :my_app, MyApp.Gettext, options
The :otp_app option (an atom representing an OTP application) has
to always be present and has to be passed to use Gettext because it's used
to determine the application to read the configuration of (:my_app in the
example above); for this reason, :otp_app can't be configured via the Mix
configuration. This option is also used to determine the application's
directory where to search messages in.
The following is a comprehensive list of supported options:
	:priv - a string representing a directory where messages will be
searched. The directory is relative to the directory of the application
specified by the :otp_app option. It is recommended to always have
this directory inside "priv", otherwise some features won't work as expected.
By default it's "priv/gettext".

	:plural_forms - a module which will act as a "pluralizer". For more
information, look at the documentation for Gettext.Plural.

	:default_locale - a string which specifies the default locale to use for
the given backend.

	:split_module_by - instead of bundling all locales into a single
module, this option makes Gettext build internal modules per locale,
per domain, or both. This reduces compilation times and beam file sizes
for large projects. For example: split_module_by: [:locale, :domain].

	:split_module_compilation - control if compilation of split modules
should happen in :parallel (the default) or :serial.

	:allowed_locales - a list of locales to bundle in the backend.
Defaults to all the locales discovered in the :priv directory.
This option can be useful in development to reduce compile-time
by compiling only a subset of all available locales.

	:interpolation - the name of a module that implements the
Gettext.Interpolation behaviour. Default: Gettext.Interpolation.Default



  
    
  
  Mix tasks configuration


You can configure Gettext Mix tasks under the :gettext key in the
configuration returned by project/0 in mix.exs:
def project() do
  [app: :my_app,
   # ...
   gettext: [...]]
end
The following is a list of the supported configuration options:
	:fuzzy_threshold - the default threshold for the Jaro distance measuring
the similarity of messages. Look at the documentation for the mix gettext.merge task (Mix.Tasks.Gettext.Merge) for more information on
fuzzy messages.

	:excluded_refs_from_purging - a regex that is matched against message
references. Gettext will preserve all messages in all POT files that
have a matching reference. You can use this pattern to prevent Gettext from
removing messages that you have extracted using another tool.

	:custom_flags_to_keep - a list of custom flags that will be kept for
existing messages during a merge. Gettext always keeps the fuzzy flag.
If you want to keep the elixir-format flag, which is also commonly
used by Gettext, add it to this list. Available since v0.23.0.

	:write_reference_comments - a boolean that specifies whether reference
comments should be written when outputting PO(T) files. If this is false,
reference comments will not be written when extracting messages or merging
messages, and the ones already found in files will be discarded.

	:write_reference_line_numbers - a boolean that specifies whether file
reference comments include line numbers when outputting PO(T) files.
Defaults to true.

	:sort_by_msgid - modifies the sorting behavior. Can be either nil (the default),
:case_sensitive, or :case_insensitive.
By default or if nil, the order of existing messages in a POT file is kept and new
messages are appended to the file. If :sort_by_msgid is set to :case_sensitive,
existing and new messages will be mixed and sorted alphabetically by msgid.
If set to :case_insensitive, the same applies but the sorting is case insensitive.
Note: this option also supports true and false for backwards compatibility,
but these values are deprecated as of v0.21.0.

	:on_obsolete - controls what happens when obsolete messages are found.
If :mark_as_obsolete, messages are kept and marked as obsolete.
If :delete, obsolete messages are deleted. Defaults to :delete.

	:store_previous_message_on_fuzzy_match - a boolean that controls
whether to store the previous message text in case of a fuzzy match.
Defaults to false.

Gettext.Backend behaviour
    



      
Defines a Gettext backend.

  
    
  
  Usage


A Gettext backend must use this module.
defmodule MyApp.Gettext do
  use Gettext.Backend, otp_app: :my_app
end
Using this module generates all the callbacks required by the Gettext.Backend
behaviour into the module that uses it. For more options and information,
see Gettext.
use Gettext.Backend Is a Recent Feature
Before version v0.26.0, you could only use Gettext to generate a backend.
Version v0.26.0 changes the way backends work so that now a Gettext backend
must use Gettext.Backend, while to use the functions in the backend you
will do use Gettext, backend: MyApp.Gettext.

Gettext.Interpolation behaviour
    



      
Behaviour to provide Gettext string interpolation.
By default, Gettext uses Gettext.Interpolation.Default as the interpolation module.

    Gettext.Interpolation.Default - gettext v0.26.2
    
    

    


  
  

    
Gettext.Interpolation.Default 
    



      
Default implementation for the Gettext.Interpolation behaviour.
Replaces %{binding_name} with the string value of the binding_name binding.

Gettext.Macros 
    



      
Macros used by Gettext to provide the gettext family of functions.
Available since v0.26.0.
Macros enable users to use gettext and get automatic extraction of translations.
See Gettext for more information.
The macros in this module that don't end with _with_backend are imported
every time you call:
use Gettext, backend: MyApp.Gettext

  
    
  
  Explicit backend


If you need to use the macros here with an explicit backend and you want extraction
to work, you can use the _with_backend versions of the macros in this module explicitly
instead.
defmodule MyApp.Gettext do
  use Gettext, otp_app: :my_app
end

defmodule MyApp.Controller do
  require Gettext.Macros

  def index(conn, _params) do
    Gettext.Macros.gettext_with_backend(MyApp.Gettext, "Hello, world!")
  end
end

Gettext.Plural behaviour
    



      
Behaviour and default implementation for finding plural forms in given
locales.
This module both defines the Gettext.Plural behaviour and provides a default
implementation for it.

  
    
  
  Plural Forms


For a given language, there is a grammatical rule on how to change words
depending on the number qualifying the word. Different languages can have
different rules.
[source]

Such grammatical rules define a number of plural forms. For example,
English has two plural forms: one for when there is just one element (the
singular) and another one for when there are zero or more than one elements
(the plural). There are languages which only have one plural form and there
are languages which have more than two.
In GNU Gettext (and in Gettext for Elixir), plural forms are represented by
increasing 0-indexed integers. For example, in English 0 means singular and
1 means plural.
The goal of this module is to determine, given a locale:
	how many plural forms exist in that locale (nplurals/1);

	to what plural form a given number of elements belongs to in that locale
(plural/2).



  
    
  
  Default Implementation


Gettext.Plural provides a default implementation of a plural module. Most
common languages used on Earth should be covered by this default implementation. If
custom pluralization rules are needed (for example, to add additional
languages) a different plural module can be specified when creating a Gettext
backend. For example, pluralization rules for the Elvish language could be
added as follows:
defmodule MyApp.Plural do
  @behaviour Gettext.Plural

  def nplurals("elv"), do: 3

  def plural("elv", 0), do: 0
  def plural("elv", 1), do: 1
  def plural("elv", _), do: 2

  # Fall back to Gettext.Plural
  defdelegate nplurals(locale), to: Gettext.Plural
  defdelegate plural(locale, n), to: Gettext.Plural
end
The mathematical expressions used in this module to determine the plural form
of a given number of elements are taken from this
page
as well as from Mozilla's guide on "Localization and
plurals".

  
    
  
  Changing Implementations


Once you have defined your custom plural forms module, you can use it
in two ways. You can set it for all Gettext backends in your
configuration:
# For example, in config/config.exs
config :gettext, :plural_forms, MyApp.Plural
or you can set it for each specific backend when you call use Gettext:
defmodule MyApp.Gettext do
  use Gettext.Backend,
    otp_app: :my_app,
    plural_forms: MyApp.Plural
end
Compile-time Configuration
Set :plural_forms in your config/config.exs and
not in config/runtime.exs, as Gettext reads this option when
compiling your backends.

Task such as mix gettext.merge use the plural
backend configured under the :gettext application, so in general
the global configuration approach is preferred.
Some tasks also allow the number of plural forms to be given
explicitly, for example:
mix gettext.merge priv/gettext --locale=gsw_CH --plural-forms=2

  
    
  
  Unknown Locales


Trying to call Gettext.Plural functions with unknown locales will result in
a Gettext.Plural.UnknownLocaleError exception.

  
    
  
  Language and Territory


Often, a locale is composed as a language and territory pair, such as
en_US. The default implementation for Gettext.Plural handles xx_YY by
forwarding it to xx (except for just Brazilian Portuguese, pt_BR, which
is not forwarded to pt as pluralization rules differ slightly). We treat the
underscore as a separator according to
ISO 15897. Sometimes, a dash - is
used as a separator (for example BCP47
locales use this as in en-US): this is not forwarded to en in the default
Gettext.Plural (and it will raise an Gettext.Plural.UnknownLocaleError exception
if there are no messages for en-US). We recommend defining a custom plural forms
module that replaces - with _ if needed.

  
    
  
  Examples


An example of the plural form of a given number of elements in the Polish
language:
iex> Gettext.Plural.plural("pl", 1)
0
iex> Gettext.Plural.plural("pl", 2)
1
iex> Gettext.Plural.plural("pl", 5)
2
iex> Gettext.Plural.plural("pl", 112)
2
As expected, nplurals/1 returns the possible number of plural forms:
iex> Gettext.Plural.nplurals("pl")
3

Gettext.Error exception
    



      
A generic error raised for a variety of possible Gettext-related reasons.

Gettext.MissingBindingsError exception
    



      
An error message raised for missing bindings errors.

Gettext.Plural.UnknownLocaleError exception
    



      
Raised when a pluralized module doesn't know how to handle a locale.

  
    
  
  Examples


raise Gettext.Plural.UnknownLocaleError, "en-US"

Gettext.PluralFormError exception
    



      
An generic error for when a plural form is missing for a given locale.

mix gettext.extract 
    



      
Extracts messages by recompiling the Elixir source code.
mix gettext.extract [OPTIONS]

messages are extracted into POT (Portable Object Template) files with a
.pot extension. The location of these files is determined by the :otp_app
and :priv options given by Gettext modules when they call use Gettext. One
POT file is generated for each message domain.
All automatically-extracted messages are assigned the elixir-autogen flag.
If a message from the POT is no longer present and has the elixir-autogen
flag, the message is removed.
Before v0.19.0, the elixir-format flag was used to detect automatically
extracted messages. This has been deprecated in v0.19.0. When extracting
with the newest version, the new elixir-autogen flag is added to all
automatically extracted messages.
All messages are assigned a format flag. When using the default
interpolation module, that flag is elixir-format. With other interpolation
modules, the flag name is defined by that implementation (see
Gettext.Interpolation.message_format/0).
If you would like to verify that your POT files are up to date with the
current state of the codebase, you can provide the --check-up-to-date
flag. This is particularly useful for automated checks and in CI systems.
This validation will fail even when the same calls to Gettext
only change location in the codebase:
mix gettext.extract --check-up-to-date

It is possible to pass the --merge option to perform merging
for every Gettext backend updated during merge:
mix gettext.extract --merge

All other options passed to gettext.extract are forwarded to the
gettext.merge task (Mix.Tasks.Gettext.Merge), which is called internally
by this task. For example:
mix gettext.extract --merge --no-fuzzy

mix gettext.merge 
    



      
Merges PO/POT files with PO files.
This task is used when messages in the source code change: when they do,
mix gettext.extract is usually used to extract the new messages to POT
files. At this point, developers or translators can use this task to "sync"
the newly-updated POT files with the existing locale-specific PO files. All
the metadata for each message (like position in the source code, comments,
and so on) is taken from the newly-updated POT file; the only things taken
from the PO file are the actual translated strings.
Fuzzy Matching
Messages in the updated PO/POT file that have an exact match (a
message with the same msgid) in the old PO file are merged as described
above. When a message in the updated PO/POT files has no match in the old
PO file, Gettext attemps a fuzzy match for that message. For example, imagine
we have this POT file:
msgid "hello, world!"
msgstr ""
and we merge it with this PO file:
# No exclamation point here in the msgid
msgid "hello, world"
msgstr "ciao, mondo"
Since the two messages are similar, Gettext takes the msgstr from the
existing message over to the new message, which it however
marks as fuzzy:
#, fuzzy
msgid "hello, world!"
msgstr "ciao, mondo"
Generally, a fuzzy flag calls for review from a translator.
Fuzzy matching can be configured (for example, the threshold for message
similarity can be tweaked) or disabled entirely. Look at the
"Options" section.

  
    
  
  Usage


mix gettext.merge OLD_FILE UPDATED_FILE [OPTIONS]
mix gettext.merge DIR [OPTIONS]

If two files are given as arguments, OLD_FILE must be a .po file and
UPDATE_FILE must be a .po/.pot file. The first one is the old PO file,
while the second one is the last generated one. They are merged and written
over the first file. For example:
mix gettext.merge priv/gettext/en/LC_MESSAGES/default.po priv/gettext/default.pot

If only one argument is given, then that argument must be a directory
containing Gettext messages (with .pot files at the root level alongside
locale directories - this is usually a "backend" directory used by a Gettext
backend, see Gettext.Backend). For example:
mix gettext.merge priv/gettext

If the --locale LOCALE option is given, then only the PO files in
<DIR>/<LOCALE>/LC_MESSAGES will be merged with the POT files in DIR. If no
options are given, then all the PO files for all locales under DIR are
merged with the POT files in DIR.

  
    
  
  Plural Forms


By default, Gettext will determine the number of plural forms for newly-generated messages
by checking the value of nplurals in the Plural-Forms header in the existing .po file. If
a .po file doesn't already exist and Gettext is creating a new one or if the Plural-Forms
header is not in the .po file, Gettext will use the number of plural forms that
the plural module (see Gettext.Plural) returns for the locale of the file being created.
The content of the Plural-Forms header can be forced through the --plural-forms-header
option (see below).

  
    
  
  Options


	--locale - a string representing a locale. If this is provided, then only the PO
files in <DIR>/<LOCALE>/LC_MESSAGES will be merged with the POT files in DIR. This
option can only be given when a single argument is passed to the task
(a directory).

	--no-fuzzy - don't perform fuzzy matching when merging files.

	--fuzzy-threshold - a float between 0 and 1 which represents the
minimum Jaro distance needed for two messages to be considered a fuzzy
match. Overrides the global :fuzzy_threshold option (see the docs for
Gettext for more information on this option).

	--plural-forms - (deprecated in v0.22.0) an integer strictly greater than 0.
If this is passed, new messages in the target PO files will have this number of empty
plural forms. This is deprecated in favor of passing the --plural-forms-header,
which contains the whole plural-forms specification. See the "Plural forms" section above.

	--plural-forms-header - the content of the Plural-Forms header as a string.
If this is passed, new messages in the target PO files will use this content
to determine the number of plurals. See the "Plural Forms" section.

	--on-obsolete - controls what happens when obsolete messages are found.
If mark_as_obsolete, messages are kept and marked as obsolete.
If delete, obsolete messages are deleted. Defaults to delete.

	--store-previous-message-on-fuzzy-match - controls if the previous
messages are recorded on fuzzy matches. Is off by default.

